How can we compute the ideal of the twisted cubic?

Chris Peterson and Hirotachi Abo

Let $k = \mathbb{Q}$. Consider a polynomial map φ from $\mathbb{A}^1_k$ to $\mathbb{A}^3_k$ defined by

$$\varphi(t) = (t^3, t^2, t).$$

Here we describe $V = \varphi\left(\mathbb{A}^1_k\right)$. A natural question is maybe:

Question 1. Is $V \subseteq \mathbb{A}^3_k$ an affine variety?

To answer to this question, it suffices to find an ideal I in $k[x, y, z]$ such that $V = V(I)$. Recall that the map φ induces a ring homomorphism $\tilde{\varphi}$ from $k[x, y, z]$ to $k[t]$. It is obvious to see that the homomorphism is surjective. Consider the kernel $J := \mathrm{Ker}(\varphi)$ of this homomorphism. The kernel is, needless to say, an ideal of $k[x, y, z]$. Thus $\tilde{\varphi}$ induces a ring homomorphism from $k[x, y, z]/J$ to $k[t]$. (This homomorphism is an isomorphism, which implies that the corresponding varieties $V(J)$ and $\mathbb{A}^1_k$ are isomorphic.) The kernel J can be computed with `Macaulay2`:

```
i1 : KK=QQ

o1 = QQ

o1 : Ring

  --  the class of all rational numbers

i2 : R=KK[x,y,z]

o2 = R

o2 : PolynomialRing

i3 : S=KK[t]

o3 = S

o3 : PolynomialRing
```

```
i4 : phi=map(S,R,{t^3,t^2,t})

                 3   2
o4 = map(S,R,{t  , t  , t})

o4 : RingMap S <--- R

i5 : J=kernel phi

                 2         3
o5 = ideal (- z  + y, - z  + x)

o5 : Ideal of R
```

So we would like to ask:

Question 2. Are V and $V(J)$ the same?

Note that if f is in J, then f vanishes at any point of V. This implies that V is contained in $V(J)$. Let $P = (a, b, c)$ be an arbitrary point of $V(J)$ (a, b and c should be in $\mathbb{Q}$). From o5 it follows that there are the following relations among a, b and c:

$$a = c^3 \text{ and } b = c^2.$$

So the coordinates of P can be written by (c^3, c^2, c). Since $c \in \mathbb{Q}$, the point P lies in V. This implies that $V = V(J)$ (the answer to Question 2), and hence V is an affine variety in $\mathbb{A}^3_k$ (the answer to Question 1).

Next we present another way to compute the ideal of the affine variety obtained as the image of the polynomial map φ. First of all, we describe φ in another way. Consider the following affine variety:

$$\Gamma_\varphi := \left\{(t, x, y, z) \in \mathbb{A}^1_k \times \mathbb{A}^3_k \mid x = t^3, y = t^2, z = t\right\}.$$

This is called the *graph* of the map φ. Then we have two polynomial maps: $\iota : \mathbb{A}^1_k \to \mathbb{A}^1_k \times \mathbb{A}^3_k$ and $\pi : \mathbb{A}^1_k \times \mathbb{A}^3_k \to \mathbb{A}^3_k$ defined by $\iota(t) = (t, t^3, t^2, t)$ and $\pi(t, x, y, z) = (x, y, z)$ respectively (see Problem 1 in Problem Set 8). Note that $\varphi = \pi \circ \iota$. Consider the following system of polynomial equations:

$$\begin{cases} x = t^3 \\ y = t^2 \\ z = t. \end{cases}$$

Solving this system for x, y and z gives us a new system of polynomial equations with variables x, y and z. The solution set of this system contains V. In other words, $V \subseteq V(I_1)$, where $I_1 = (x - t^3, y - t^2, z - t) \cap k[x, y, z]$. This ideal I_1 can be computed as follows:

```
i6 : SxR=KK[t,x,y,z,MonomialOrder=>Lex]

o6 = SxR

o6 : PolynomialRing
```

Remark. Choose the lexicographic ordering (the default is the graded reverse lexicographic ordering).

Define the ideal gamma of the graph Γ_φ:

```
i7 : gamma=ideal(x-t^3,y-t^2,z-t)

                3         2
o7 = ideal (- t  + x, - t  + y, - t + z)

o7 : Ideal of SxR
```

Compute the Gröbner basis for gamma (the definition of Gröbner basis will be introduced soon):

```
i8 : gb gamma

o8 = | t-z y-z2 x-z3 |

o8 : GroebnerBasis
```

In the Gröbner basis, we can find the polynomials containing only the variables x, y and z. Select these polynomials:

```
i9 : elim=ideal selectInSubring(1,gens gb gamma)

                  2       3
o9 = ideal (y - z , x - z )

o9 : Ideal of SxR
```

The ideal elim is still in the coordinate ring of $\mathbb{A}^1_k \times \mathbb{A}^3_k$. Substitute elim into the coordinate ring R of $\mathbb{A}^3_k$:

```
i10 : elim'=substitute(elim,R)

                   2          3
o10 = ideal (- z  + y, - z  + x)

o10 : Ideal of R
```

Output `o10` is the desired ideal I_1. Now it remains only to show that the solutions of the system

$$\begin{cases} x = z^3 \\ y = z^2 \end{cases}$$

can be extended the original system of polynomial equations. However, this statement can be shown by the same argument we discussed previously, so we omitte this.

Remark. The curve in $\mathbb{A}^3_k$ obtained as the image of φ, or equivalently the variety in $\mathbb{A}^3_k$ defined by the ideal $(x - z^3, y - z^2)$, is called the *twisted cubic*.